
FIGURE 9. 1
A process statement block diagram.


FIGURE 9. 2
A process runs in zero time, repeats forever, unless suspended.


ARCHITECTURE sequentiality_demo OF partial_process IS


BEGIN


   PROCESS


   BEGIN


     ...


     x <= a;


     y <= b;


     ...


   END PROCESS;


END sequentiality_demo;
FIGURE 9. 3
Sequentiality in process statements.  The amount of real time between the execution of one statement and the next is zero.  Both statements occur in one simulation cycle.


ARCHITECTURE execution_time_demo OF partial_process IS


BEGIN


   PROCESS


   BEGIN


     ...


     x <= a AFTER 10 NS;


     y <= b AFTER 6 NS;


     ...


   END PROCESS;


END execution_time_demo;

FIGURE 9. 4
Partial code for demonstration of zero execution time of a process statement.


ARCHITECTURE data_availability_demo OF partial_process IS


BEGIN


   PROCESS


   BEGIN


      ...


      x <= '1';


      IF x = '1' THEN


         Perform_action_1


      ELSE


         Perform_action_2

      END IF;


      ...


   END PROCESS;


END data_availability_demo;

FIGURE 9. 5
Partial code for demonstrating delay in assignment of values to signals.



ARCHITECTURE …


ARCHITECTURE …



BEGIN




BEGIN



     …




     …



     a <= b;



     PROCESS (b)



     …




       …



     c <= d;



        a <= b;



     …




     END PROCESS;



END …;



     …








     c <= d;








     …








END …;

(a) (b)

FIGURE 9. 6
A simple process with sensitivity list,  (a) signal assignment,  (b) equivalent process statement.

FIGURE 9. 7
A positive edge trigger D-Type flip-flop with asynchronous set and reset inputs.


ENTITY d_sr_flipflop IS 


   GENERIC (sq_delay, rq_delay, cq_delay : TIME := 6 NS); 


   PORT (d, set, rst, clk : IN BIT; q, qb : OUT BIT); 


END d_sr_flipflop;


 --


ARCHITECTURE behavioral OF d_sr_flipflop IS


   SIGNAL state : BIT := '0';


BEGIN


   dff: PROCESS (rst, set, clk)


   BEGIN


      IF set = '1' THEN


         state <= '1' AFTER sq_delay;


      ELSIF rst = '1' THEN


         state <= '0' AFTER rq_delay;


      ELSIF clk = '1' AND clk'EVENT THEN


         state <= d AFTER cq_delay; 


      END IF;


   END PROCESS dff;


   q <= state;


   qb <= NOT state;


END behavioral;

FIGURE 9. 8
VHDL description for the flip-flop of Figure 9.7.


ARCHITECTURE average_delay_behavioral OF d_sr_flipflop IS


BEGIN


   dff: PROCESS (rst, set, clk)


      VARIABLE state : BIT := '0';


   BEGIN


      IF set = '1' THEN


         state := '1';


      ELSIF rst = '1' THEN


         state := '0';


      ELSIF clk = '1' AND clk'EVENT THEN


         state := d;


      END IF;


      q <= state AFTER (sq_delay + rq_delay + cq_delay) /3;


      qb <= NOT state AFTER (sq_delay + rq_delay + cq_delay) /3;


   END PROCESS dff;


END average_delay_behavioral;

FIGURE 9. 9
Alternative architecture for d_sr_flipflop entity; reducing ( delay By One.

FIGURE 9. 10
Simultanious Simulation of behavioral and average_delay_behavioral Architectures of d_sr_flipflop.  All Events Are Observed.

ENTITY d_sr_flipflop IS

  GENERIC (sq_delay, rq_delay, cq_delay : TIME := 6 NS);

  PORT (d, set, rst, clk : IN BIT; q, qb : OUT BIT);

END ENTITY;

--

ARCHITECTURE behavioral OF d_sr_flipflop IS

BEGIN

   dff: PROCESS (rst, set, clk)

      TYPE bit_time IS RECORD

         state : BIT;  delay : TIME;

      END RECORD;

      VARIABLE sd : bit_time := ('0', 0 NS);

   BEGIN

      IF set = '1' THEN

         sd := ('1', sq_delay);

      ELSIF rst = '1' THEN

         sd := ('0', rq_delay);

      ELSIF clk = '1' AND clk'EVENT THEN

         sd := (d, cq_delay);

      END IF;

      q <= sd.state AFTER sd.delay;

      qb <= NOT sd.state AFTER sd.delay;

   END PROCESS dff;

END behavioral;

FIGURE 9. 11
A flipflop using state-delay record.

FIGURE 9. 12
Syntax Details of a Process Statement With Sensitivity List, Declarative Part, and Statement Part.


FIGURE 9. 13
Activation of a postponed process.

PACKAGE bt IS

   TYPE bit_time IS RECORD

      state : BIT;  delay : TIME;

   END RECORD;

   SHARED VARIABLE sd : bit_time := ('0', 0 NS);

END PACKAGE bt;

--

USE WORK.bt.ALL;

ENTITY d_sr_flipflop IS

   GENERIC (sq_delay, rq_delay, cq_delay : TIME := 6 NS);

   PORT (d, set, rst, clk : IN BIT; q, qb : OUT BIT);

BEGIN

   dff: PROCESS (rst, set, clk)

   BEGIN

      IF set = '1' THEN

         sd := ('1', sq_delay);

      ELSIF rst = '1' THEN

         sd := ('0', rq_delay);

      ELSIF clk = '1' AND clk'EVENT THEN

         sd := (d, cq_delay);

      END IF;

   END PROCESS dff;

END ENTITY;

--

ARCHITECTURE behavioral OF d_sr_flipflop IS

BEGIN

   dff_arch: PROCESS (rst, set, clk)

   BEGIN

      q <= sd.state AFTER sd.delay;

      qb <= NOT sd.state AFTER sd.delay;

   END PROCESS dff_arch;

END behavioral;

FIGURE 9. 14
A passive process statement may appear in the entity statement part.

long_runing : LOOP

   . . .

   IF x = 25 THEN EXIT;

   END IF;

   . . .

END LOOP long_runing;

FIGURE 9. 15
Partial Code for Demonstrating Exiting from a Potentially Infinite Loop.


loop_1 : FOR i IN 5 TO 25 LOOP


   . . .


   sequential_statement_1;


   . . .


   sequential_statement_2;


   . . .


   loop_2 : WHILE j <= 90 LOOP


   . . .


      sequential_statement_3;


      sequential_statement_4;


      . . .


      NEXT loop_1 WHEN condition_1;


      . . .


      sequential_statement_5;


      sequential_statement_6;


   . . .


   END LOOP loop_2;


   . . .


END LOOP loop_1;

FIGURE 9. 16
Partial Code for Demonstrating Conditional Next Statements in a Loop.


ARCHITECTURE behavioral OF d_sr_flipflop IS


   SIGNAL state : BIT := '0';


BEGIN


   dff: PROCESS (rst, set, clk)


   BEGIN


      ASSERT 


         (NOT (set = '1' AND rst = '1')) 


      REPORT 


         "set and rst are both 1" 


      SEVERITY NOTE; 


      IF set = '1' THEN


         state <= '1' AFTER sq_delay;


      ELSIF rst = '1' THEN


         state <= '0' AFTER rq_delay;


      ELSIF clk = '1' AND clk'EVENT THEN


         state <= d AFTER cq_delay; 


      END IF;


   END PROCESS dff;


   q <= state;


   qb <= NOT state;


END behavioral;

FIGURE 9. 17
A Modified behavioral Architecture of d_sr_flipflop, Checking for Simultaneous Assertion of set and rst.

FIGURE 9. 18
Setup and Hold Times for A Positive Edge Trigger D-Type Flip-Flop.


ENTITY d_sr_flipflop IS


   GENERIC (sq_delay, rq_delay, cq_delay : TIME := 6 NS;


           setup, hold : TIME := 4 NS); 


   PORT (d, set, rst, clk : IN BIT; q, qb : OUT BIT); 


BEGIN


   ASSERT 


      (NOT (clk = '1' AND clk'EVENT AND NOT d'STABLE(setup) ))


   REPORT 


      "setup time violation" 


   SEVERITY WARNING;


   ASSERT 


      (NOT (d'EVENT AND clk = '1' AND NOT clk'STABLE(hold) ))


   REPORT 


      "Hold time violation" 


   SEVERITY WARNING;


END d_sr_flipflop;


 --


ARCHITECTURE behavioral OF d_sr_flipflop IS


   SIGNAL state : BIT := '0';


BEGIN


   dff: PROCESS (rst, set, clk)


   BEGIN


      ASSERT 


         (NOT (set = '1' AND rst = '1')) 


      REPORT 


         "set and rst are both 1" 


      SEVERITY NOTE; 


      IF set = '1' THEN


         state <= '1' AFTER sq_delay;


      ELSIF rst = '1' THEN


         state <= '0' AFTER rq_delay;


      ELSIF clk = '1' AND clk'EVENT THEN


         state <= d AFTER cq_delay; 


      END IF;


   END PROCESS dff;


   q <= state;


   qb <= NOT state;


END behavioral;

FIGURE 9. 19
A Complete D-Flip-Flop Description, Using Assertion Statements for Illegal Set-Reset Combinations, and Setup and Hold Time Violations.

FIGURE 9. 20
A Moore Machine State Diagram for Detecting 1011 Sequence.


ENTITY moore_detector IS 


   PORT (x, clk : IN BIT; z : OUT BIT);


END moore_detector;


   --


ARCHITECTURE behavioral_state_machine OF moore_detector IS


   TYPE state IS (reset, got1, got10, got101, got1011);


   SIGNAL current : state := reset;


BEGIN


   PROCESS


   BEGIN


      CASE current IS 


      WHEN reset => 


         WAIT UNTIL clk = '1'; 


          IF x = '1' THEN current <= got1;


          ELSE current <= reset;


          END IF;


      WHEN got1 => 


          WAIT UNTIL clk = '1';


          IF x = '0' THEN current <= got10;


          ELSE current <= got1;


          END IF;


      WHEN got10 =>


          WAIT UNTIL clk = '1';


          IF x = '1' THEN current <= got101;


          ELSE current <= reset;


          END IF;


      WHEN got101 =>


          WAIT UNTIL clk = '1';


          IF x = '1' THEN current <= got1011;


          ELSE current <= got10;


          END IF;


      WHEN got1011 =>


          z <= '1';


          WAIT UNTIL clk = '1';


          IF x = '1' THEN current <= got1;


          ELSE current <= got10;


          END IF;


      END CASE;


      WAIT FOR 1 NS;


      z <= '0';


   END PROCESS;


END behavioral_state_machine;

FIGURE 9. 21
VHDL Description of the 1011 Sequence Detector of Figure 9.20, Using Process and Wait Statements.


ENTITY moore_detector IS


   PORT (x, clk : IN BIT; z : OUT BIT);


END moore_detector;


--


ARCHITECTURE behavioral_state_machine OF moore_detector IS


   TYPE state IS (reset, got1, got10, got101, got1011);


   SIGNAL current : state := reset;


BEGIN


   PROCESS (clk)


   BEGIN


      IF clk = '1' THEN


         CASE current IS


         WHEN reset =>


            IF x = '1' THEN current <= got1;


            ELSE current <= reset;


            END IF;


         WHEN got1 => 


            IF x = '0' THEN current <= got10;


            ELSE current <= got1;


            END IF;


         WHEN got10 =>


            IF x = '1' THEN current <= got101;


            ELSE current <= reset;


            END IF;


         WHEN got101 =>


            IF x = '1' THEN current <= got1011;


            ELSE current <= got10;


            END IF;


         WHEN got1011 =>


            IF x = '1' THEN current <= got1;


            ELSE current <= got10;


            END IF;


         END CASE;


      END IF;


   END PROCESS;


   z <= '1' WHEN current = got1011 ELSE '0';


END behavioral_state_machine;
FIGURE 9. 22
A simple state machine description.


FIGURE 9. 23
Mealy machine detecting 101.

ENTITY asynch_reset_detector IS

  PORT (x, r, clk : IN BIT; z : OUT BIT);

END ENTITY;

--

ARCHITECTURE behavioral OF asynch_reset_detector IS

   TYPE state IS (a, b, c);
  

   SIGNAL nxt, present : state;

BEGIN

   reg : PROCESS (clk, r)

   BEGIN

      IF r = '1' THEN       

         present <= a;

      ELSIF (clk'EVENT AND clk = '1') THEN
     

         present <= nxt;

      END IF;

   END PROCESS;

   --

   logic : PROCESS (present, x)

   BEGIN

      z <= '0';    

      CASE present IS      



      WHEN a => 

         IF x = '0' THEN nxt <= a;  ELSE nxt <= b; END IF;

      WHEN b =>

         IF x = '0' THEN nxt <= c;  ELSE nxt <= b; END IF;

      WHEN c =>       

         IF x = '0' THEN nxt <= a;  ELSE nxt <= b; END IF;

      END CASE;

      IF present = c AND x = '1' THEN z <= '1'; END IF;

   END PROCESS;

END behavioral;

FIGURE 9. 24
VHDL description for a state machine with asynchronous reset.


...


phase2: PROCESS 


BEGIN


   WAIT UNTIL c1 = '0';


   WAIT FOR 10 NS;


   c2 <= '1';


   WAIT FOR 480 NS;


   c2 <= '0';


END PROCESS phase2;


...
FIGURE 9. 25
Partial Code for Generation of Second Phase of a Two Phase Non-overlapping Clocking.

FIGURE 9. 26
Two Non-overlapping Phases of Clock, c2 Generated by phase2 Process of Figure 9.25.

FIGURE 9. 27
Signals of a Fully Responsive two-line Handshaking.

System A:


       -- start the following when ready to send


       data_lines <= newly_prepared_data;


       data_ready <= '1';


       WAIT UNTIL accepted = '1';


       data_ready <= '0';


       --can use data_lines for other purposes

System B:
       


       -- start the following when ready to accept data


       WAIT UNTIL data_ready = '1';


       accepted <= '1';


       -- start processing the newly received data


       WAIT UNTIL data_ready = '0';


       accepted <= '0';

FIGURE 9. 28
VHDL Code for Fully Responsive Two Line Handshaking.


FIGURE 9. 29
Interfacing system_a and system_b.  System_i Uses Handshaking to Talk to Both Systems.


ENTITY system_i IS


   PORT (in_data : IN BIT_VECTOR (3 DOWNTO 0);


                out_data : OUT BIT_VECTOR (15 DOWNTO 0);


                in_ready, out_received : IN BIT; in_received, out_ready : OUT BIT);


END system_i;


  --


ARCHITECTURE waiting OF system_i IS


   SIGNAL buffer_full, buffer_picked : BIT := '0';


   SIGNAL word_buffer : BIT_VECTOR (15 DOWNTO 0);


BEGIN


   a_talk: PROCESS


      VARIABLE count : INTEGER RANGE 0 TO 4 := 0;


   BEGIN


      WAIT UNTIL in_ready = '1';


      count := count + 1;


      CASE count IS


      WHEN 0 => NULL;


      WHEN 1 => word_buffer (03 DOWNTO 00) <= in_data;


      WHEN 2 => word_buffer (07 DOWNTO 04) <= in_data;


      WHEN 3 => word_buffer (11 DOWNTO 08) <= in_data;


      WHEN 4 => word_buffer (15 DOWNTO 12) <= in_data;


                  buffer_full <= '1';


                  WAIT UNTIL buffer_picked = '1';


                  buffer_full <= '0';


                  count := 0;


      END CASE;


      in_received <= '1';


      WAIT UNTIL in_ready = '0';


      in_received <= '0';


   END PROCESS a_talk;


   b_talk: PROCESS


   BEGIN


      IF buffer_full = '0' THEN WAIT UNTIL buffer_full = '1'; END IF;


      out_data <= word_buffer;


      buffer_picked <= '1';


      WAIT UNTIL buffer_full = '0';


      buffer_picked <= '0';


      out_ready <= '1';


      WAIT UNTIL out_received = '1';


      out_ready <= '0';


   END PROCESS b_talk;


END waiting;

FIGURE 9. 30
VHDL Model for the Interface Between Systems A and B of Figure 9.29.


FIGURE 9. 31
Bus arbiter interface.

ENTITY arbiter IS

   PORT (request : IN BIT_VECTOR (3 DOWNTO 0); 

             grant : BUFFER BIT_VECTOR (3 DOWNTO 0); clock : IN BIT);

END arbiter;

--

ARCHITECTURE behavioral OF arbiter IS

BEGIN

   wait_cycle: PROCESS 

   BEGIN

      IF clock = '0' THEN

         WAIT FOR 20 NS;

         FOR i IN request'RANGE LOOP

            IF request(i) = '1' THEN

               grant <= "0000"; grant (i) <= '1';

            ELSE

               grant (i) <= '0';

            END IF; 

         END LOOP;

      END IF;

      WAIT ON clock;

   END PROCESS wait_cycle;

END behavioral;

FIGURE 9. 32
Bus arbiter description.

ENTITY arbtest IS END arbtest;

--

ARCHITECTURE io OF arbtest IS

   SIGNAL clk : BIT;

   SIGNAL r, g : BIT_VECTOR (3 DOWNTO 0);

   CONSTANT t : TIME := 1 US;

   TYPE time_array IS ARRAY (3 DOWNTO 0) OF TIME;

   CONSTANT delays : time_array := (4 US, 3 US, 15 US, 8 US);

BEGIN

   arb : ENTITY WORK.arbiter PORT MAP (r, g, clk);

   clk <= NOT clk AFTER t / 2 WHEN NOW < 40 US ELSE clk;

   sources : FOR i IN  r'RANGE GENERATE

      PROCESS

      BEGIN

         WAIT FOR delays (i);

         r(i) <= '1';

         WAIT UNTIL g(i) = '1';

         WAIT UNTIL clk = '0';

         r(i) <= '0';

      END PROCESS;

   END GENERATE;

END io;

FIGURE 9. 33
Testing the arbiter.


FIGURE 9. 34
Serial_to_parallel interface.


FIGURE 9. 35
RS232 frame.

ENTITY serial2parallel IS

   GENERIC (bps : INTEGER);

   PORT (serial, received : IN qit; dataready : BUFFER qit; 

             overrun, frame_error : OUT qit; 

             parallel_out : BUFFER qit_vector (7 DOWNTO 0));

END serial2parallel;

--

ARCHITECTURE waiting OF serial2parallel IS

BEGIN

   collect : PROCESS  

      VARIABLE buff : qit_vector (7 DOWNTO 0);

      CONSTANT half_bit : TIME := (1000000.0/REAL(bps))/2.0 * 1 US;

      CONSTANT full_bit : TIME := (1000000.0/REAL(bps)) * 1 US;

   BEGIN

      WAIT UNTIL serial = '0';

      WAIT FOR half_bit; 

      FOR count IN 0 TO 7 LOOP

         WAIT FOR full_bit;

         buff (count) := serial;

      END LOOP;

      WAIT FOR full_bit;

      IF serial = '0' THEN

         frame_error <= '1';

         WAIT UNTIL serial = '1';

      ELSE

         frame_error <= '0';

         dataready <= '1';

         parallel_out <= buff;

         WAIT UNTIL received = '1';

         WAIT UNTIL received = '0';

         dataready <= '0';

      END IF;

   END PROCESS collect;

   --

   too_fast : PROCESS

   BEGIN

      IF dataready = '1' THEN

         WAIT UNTIL serial = '0';

         IF dataready = '1' THEN

            overrun <= '1';

         END IF;

      ELSE

         overrun <= '0';

      END IF;

      WAIT ON dataready;

   END PROCESS too_fast;

END waiting;

FIGURE 9. 36
Serial2parallel VHDL description.


USE STD.TEXTIO.ALL;


  ...


TYPE state IS (reset, got1, got10, got101);


TYPE state_vector IS ARRAY (NATURAL RANGE <>) OF state;


FUNCTION one_of (sources : state_vector) RETURN state IS


   VARIABLE l : LINE;


   FILE flush : TEXT OPEN WRITE_MODE IS "/dev/tty"; 


BEGIN


   FOR i IN sources'RANGE LOOP


      WRITE (l, state'IMAGE(sources(i)), LEFT, 7);


   END LOOP;


   WRITELINE (flush, l);


   RETURN sources(sources'LEFT);


END one_of;

FIGURE 9. 37
A Resolution Function that Writes Its Active Drivers Each Time It Is Called.

PROCEDURE display (SIGNAL value1, value2 : BIT) IS

   FILE flush : TEXT OPEN WRITE_MODE IS "/dev/tty"; 

   VARIABLE filler : STRING (1 TO 3) := " ..";

   VARIABLE l : LINE;

BEGIN

   WRITE (l, NOW, RIGHT, 8, NS);

   IF value1'EVENT AND value2'EVENT THEN

      WRITE (l, value1, RIGHT, 3);

      WRITE (l, value2, RIGHT, 3);

   ELSIF value1'EVENT THEN

      WRITE (l, value1, RIGHT, 3);

      WRITE (l, filler, LEFT, 0);

   ELSE

         WRITE (l, filler, LEFT, 0);

         WRITE (l, value2, RIGHT, 3);

   END IF;

   WRITELINE (flush, l);

END display;

FIGURE 9. 38
A Display Procedure for Displaying Time and Value of a Signal that Has Just Changed.

USE STD.TEXTIO.ALL;

ENTITY two_phase_clock IS END two_phase_clock;

--

ARCHITECTURE input_output OF two_phase_clock IS

   PROCEDURE display (SIGNAL value1, value2 : BIT) IS

      FILE flush : TEXT OPEN WRITE_MODE IS "/dev/tty"; 

      VARIABLE filler : STRING (1 TO 3) := " ..";

      VARIABLE l : LINE;

   BEGIN

      WRITE (l, NOW, RIGHT, 8, NS);

      IF value1'EVENT AND value2'EVENT THEN

         WRITE (l, value1, RIGHT, 3);

         WRITE (l, value2, RIGHT, 3);

      ELSIF value1'EVENT THEN

         WRITE (l, value1, RIGHT, 3);

         WRITE (l, filler, LEFT, 0);

      ELSE

         WRITE (l, filler, LEFT, 0);

         WRITE (l, value2, RIGHT, 3);

      END IF;

      WRITELINE (flush, l);

   END display;

   SIGNAL c1 : BIT := '1';

   SIGNAL c2 : BIT := '0';

BEGIN

   phase1: c1 <= NOT c1 AFTER 500 NS WHEN NOW < 4 US ELSE c1;

   phase2: PROCESS 

   BEGIN

      WAIT UNTIL c1 = '0';

      WAIT FOR 10 NS;

      c2 <= '1';

      WAIT FOR 480 NS;

      c2 <= '0';

   END PROCESS phase2;

   display (c1, c2);

END input_output;

FIGURE 9. 39
Using display Procedure for Displaying Two Non-overlapping Clock Phases.

USE STD.TEXTIO.ALL;

USE WORK.displaying.ALL;

ENTITY two_phase_clock IS END two_phase_clock;

--

ARCHITECTURE input_output OF two_phase_clock IS

   SIGNAL c1 : BIT := '1';

   SIGNAL c2 : BIT := '0';

   FILE data : TEXT OPEN WRITE_MODE IS "clock.out";

BEGIN

   phase1: c1 <= NOT c1 AFTER 500 NS WHEN NOW < 4 US; 

   phase2: PROCESS 

   BEGIN

      WAIT UNTIL c1 = '0';

      WAIT FOR 10 NS;

      c2 <= '1';

      WAIT FOR 480 NS;

      c2 <= '0';

   END PROCESS phase2;

   display (c1, c2, data);

END input_output;

         (a)

USE STD.TEXTIO.ALL;

PACKAGE displaying IS

   PROCEDURE display (SIGNAL value1, value2 : BIT; FILE flush : TEXT);

END displaying;

--

PACKAGE BODY displaying IS

   PROCEDURE display (SIGNAL value1, value2 : BIT; FILE flush : TEXT) IS

      VARIABLE filler : STRING (1 TO 3) := " ..";

      VARIABLE l : LINE;

   BEGIN

      WRITE (l, NOW, RIGHT, 8, NS);

      IF value1'EVENT AND value2'EVENT THEN

         WRITE (l, value1, RIGHT, 3);

         WRITE (l, value2, RIGHT, 3);

      ELSIF value1'EVENT THEN

         WRITE (l, value1, RIGHT, 3);

         WRITE (l, filler, LEFT, 0);

      ELSE

         WRITE (l, filler, LEFT, 0);

         WRITE (l, value2, RIGHT, 3);

      END IF;

      WRITELINE (flush, l);

   END display;

END displaying;

         (b)

FIGURE 9. 40
The input_output Architecture of two_phase_clock Circuit with a Process Statement for Generating a Simulation Report,  (a) displaying package,  (b) architecture.

FIGURE 9. 41
File Generated by Running The input_output Architecture of Figure 9.40.

USE STD.TEXTIO.ALL;

ENTITY two_phase_clock IS END two_phase_clock;

--

ARCHITECTURE input_output OF two_phase_clock IS

   SIGNAL c1 : BIT := '1';

   SIGNAL c2 : BIT := '0';

   SIGNAL print_tick : BIT := '0';

   CONSTANT print_resolution : TIME := 5 NS;

BEGIN

   phase1: c1 <= NOT c1 AFTER 500 NS WHEN NOW < 2 US ELSE c1;

   phase2: PROCESS 

   BEGIN

      WAIT UNTIL c1 = '0';

      WAIT FOR 10 NS;

      c2 <= '1';

      WAIT FOR 480 NS;

      c2 <= '0';

   END PROCESS phase2;

   print_tick <= NOT print_tick AFTER print_resolution WHEN NOW <= 2 US ELSE UNAFFECTED;

   writing: PROCESS (print_tick, c1, c2)

      FILE flush : TEXT OPEN WRITE_MODE IS "clock4.out"; 

      VARIABLE header : STRING (1 TO 18) := "         c1   c2  ";

      VARIABLE l : LINE;

      PROCEDURE append_wave_slice (SIGNAL s : BIT) IS

         VARIABLE lo_value : STRING (1 TO 3) := "|  ";

         VARIABLE hi_value : STRING (1 TO 3) := "  |";

         VARIABLE lo_to_hi : STRING (1 TO 3) := ".-+";

         VARIABLE hi_to_lo : STRING (1 TO 3) := "+-.";

      BEGIN

         IF s'LAST_EVENT < print_resolution AND s'LAST_VALUE /= s THEN

           IF s = '1' THEN

             WRITE (l, lo_to_hi, RIGHT, 5);

           ELSE

             WRITE (l, hi_to_lo, RIGHT, 5);

           END IF;

         ELSE

           IF s = '1' THEN

             WRITE (l, hi_value, RIGHT, 5);

           ELSE

             WRITE (l, lo_value, RIGHT, 5);

           END IF;

         END IF;

      END PROCEDURE append_wave_slice;  

   BEGIN

      IF NOW = 0 US THEN 

         WRITE (l, header, LEFT, 0);

         WRITELINE (flush, l);

      END IF;

      WRITE (l, NOW, RIGHT, 8, NS);

      append_wave_slice (c1);

      append_wave_slice (c2);

      WRITELINE (flush, l);

   END PROCESS writing;

END input_output;

FIGURE 9. 42
Generating an ASCII Plot File with 5 NS Time Resolution.

                         c1   c2     

       ...            

   480 ns    |  |     

   485 ns    |  |     

   490 ns    |  |     

   495 ns    |  |     

   500 ns  +-.  |     

   505 ns  |    |     

   510 ns  |    |     

   510 ns  |    .-+   

   515 ns  |      |   

   520 ns  |      |   

   525 ns  |      |

FIGURE 9. 43
Partial Plot Generated by the ploting Process of Figure 9.42.

FIGURE 9. 44
Block Diagram of the Sequential Comparator Circuit.

FIGURE 9. 45
Partitioning Sequential Comparator Circuit into Smaller Functional Components.

FIGURE 9. 46
Standard MSI Parts for the Implementation of Sequential Comparator According to the Partitioning of Figure 9.45.


USE WORK.basic_utilities.ALL;


ENTITY ls85_comparator IS


   GENERIC (prop_delay : TIME := 10 NS);


   PORT (a, b : IN qit_vector (3 DOWNTO 0);  gt, eq, lt : IN qit;


             a_gt_b, a_eq_b, a_lt_b : OUT qit);


END ls85_comparator;


--


ARCHITECTURE behavioral OF ls85_comparator IS


BEGIN


   PROCESS (a, b, gt, eq, lt)


   BEGIN


      IF a > b THEN 


         a_gt_b <= '1' AFTER prop_delay; 


         a_eq_b <= '0' AFTER prop_delay; 


         a_lt_b <= '0' AFTER prop_delay; 


      ELSIF a < b THEN 


         a_gt_b <= '0' AFTER prop_delay; 


         a_eq_b <= '0' AFTER prop_delay; 


         a_lt_b <= '1' AFTER prop_delay; 


      ELSIF a = b THEN 


         a_gt_b <= gt AFTER prop_delay;  


         a_eq_b <= eq AFTER prop_delay;  


         a_lt_b <= lt AFTER prop_delay;  


      END IF;


   END PROCESS;


END behavioral;

FIGURE 9. 47
Behavioral Description of the 74LS85 4-Bit Magnitude Comparator.


USE WORK.basic_utilities.ALL;


ENTITY ls377_register IS


   GENERIC (prop_delay : TIME := 7 NS);


   PORT (clk, g_bar : IN qit; d8 : IN qit_vector (7 DOWNTO 0);


             q8 : OUT qit_vector (7 DOWNTO 0));


END ls377_register;


--


ARCHITECTURE dataflow OF ls377_register IS


   SIGNAL GUARD : BOOLEAN;


BEGIN


   GUARD <= NOT clk'STABLE AND clk = '1' AND (g_bar = '0');


   q8 <= GUARDED d8 AFTER prop_delay;


END dataflow;

FIGURE 9. 48
Dataflow Description of the 74LS377 8-Bit Clocked Register.

USE WORK.basic_utilities.ALL;

ENTITY ls163_counter IS

   GENERIC (prop_delay : TIME := 12 NS);

   PORT (clk, clr_bar, ld_bar, enp, ent : IN qit;

             abcd : IN qit_vector (3 DOWNTO 0); 

             q_abcd : OUT qit_vector (3 DOWNTO 0); rco : OUT qit);

END ls163_counter;

--

ARCHITECTURE behavioral OF ls163_counter IS

BEGIN

   counting : PROCESS (clk)

      VARIABLE internal_count : qit_vector (3 DOWNTO 0) := "0000";

   BEGIN

      IF (clk = '1') THEN

         IF (clr_bar = '0') THEN

            internal_count := "0000";

         ELSIF (ld_bar = '0') THEN

            internal_count := abcd;

         ELSIF (enp = '1' AND ent = '1') THEN

            internal_count := inc (internal_count);

         END IF;

         IF (internal_count = "1111" AND ent = ‘1’) THEN  

            rco <= '1' AFTER prop_delay;

         ELSE  

            rco <= '0';

         END IF;

         q_abcd <= internal_count AFTER prop_delay;

      END IF;

   END PROCESS counting;

END behavioral;

FIGURE 9. 49
Behavioral Description of the 74LS163 4-Bit Synchronous Counter.

FIGURE 9. 50
Composition Aspect of the Sequential Comparator.

USE WORK.basic_utilities.ALL;

ENTITY sequential_comparator IS 

   PORT (data_in : IN qit_vector (7 DOWNTO 0);

             clk, clear_bar, load_bar : IN qit; 

             count_in : IN qit_vector (3 DOWNTO 0);

             count : OUT qit_vector (3 DOWNTO 0) );

BEGIN

   ASSERT NOT

      ((clk='0' AND NOT clk'STABLE) AND NOT clk'DELAYED'STABLE (1 US))

   REPORT "Minimum Clock Width Violation" SEVERITY WARNING;

END sequential_comparator;

--

ARCHITECTURE structural OF sequential_comparator IS

   COMPONENT d_register 

      PORT (clk, g_bar : IN qit; d8 : IN qit_vector (7 DOWNTO 0);

                 q8 : OUT qit_vector (7 DOWNTO 0));

    END COMPONENT;

   COMPONENT comparator 

      PORT (a, b : IN qit_vector (3 DOWNTO 0);  gt, eq, lt : IN qit;

                 a_gt_b, a_eq_b, a_lt_b : OUT qit);

   END COMPONENT;

   COMPONENT counter

      PORT (clk, clr_bar, ld_bar, enp, ent : IN qit;

          abcd : IN qit_vector (3 DOWNTO 0); 

          q_abcd : OUT qit_vector (3 DOWNTO 0); rco : OUT qit);

   END COMPONENT;

   SIGNAL gnd : qit := '0'; SIGNAL vdd : qit := '1';

   SIGNAL old_data : qit_vector (7 DOWNTO 0);

   SIGNAL compare_out : qit;

   SIGNAL gt_i, eq_i, lt_i : qit;

BEGIN

   reg: d_register PORT MAP (clk, gnd, data_in, old_data);

   cmp_lo: comparator PORT MAP (data_in (3 DOWNTO 0), 

             old_data (3 DOWNTO 0), gnd, vdd, gnd, gt_i, eq_i, lt_i);

   cmp_hi: comparator PORT MAP (data_in (7 DOWNTO 4), 

             old_data (7 DOWNTO 4), gt_i, eq_i, lt_i, OPEN, compare_out, OPEN);

   cnt: counter PORT MAP (clk, clear_bar, load_bar, vdd, compare_out,

             count_in, count, OPEN);

END structural;

FIGURE 9. 51
Structural Implementation of the Sequential Comparator.


USE WORK.ALL;


CONFIGURATION standard OF sequential_comparator IS


   FOR structural


      FOR reg : d_register 


         USE ENTITY WORK.ls377_register (dataflow)


         GENERIC MAP (prop_delay => 15 NS);


      END FOR;


      FOR ALL : comparator 


         USE ENTITY WORK.ls85_comparator (behavioral)


         GENERIC MAP (prop_delay => 18 NS);


      END FOR;


      FOR cnt : counter 


         USE ENTITY WORK.ls163_counter (behavioral)


         GENERIC MAP (prop_delay => 22 NS);


      END FOR;


   END FOR;


END standard;
FIGURE 9. 52
Configuring the structural Architecture of the sequential_comparator Entity.


USE WORK.basic_utilities.ALL;


ENTITY test_sequential_comparator IS END test_sequential_comparator;


  --


ARCHITECTURE input_output OF test_sequential_comparator IS


   COMPONENT seq_comp  


   PORT (data_in : IN qit_vector (7 DOWNTO 0);


             clk, clear_bar, load_bar : IN qit; 


             count_in : IN qit_vector (3 DOWNTO 0);


             count : OUT qit_vector (3 DOWNTO 0) );


   END COMPONENT;


   FOR mfi : seq_comp USE CONFIGURATION WORK.standard;


   SIGNAL data : qit_vector (7 DOWNTO 0);


   SIGNAL ck, cl_bar, ld_bar : qit;


   SIGNAL cnt : qit_vector (3 DOWNTO 0);


   SIGNAL cnt_out : qit_vector (3 DOWNTO 0);


BEGIN


   ck <= NOT ck AFTER 2 US WHEN NOW <= 70 US ELSE ck;


   cl_bar <= '1', '0' AFTER 60 US;


   ld_bar <= '1', '0' AFTER 50 US, '1' AFTER 55 US;


   cnt <= "1111", "1011" AFTER 40 US, "0111" AFTER 55 US;


   data <= "00000000", "01110111" AFTER 3 US, "10101100" AFTER 5 US, 


             "01010100" AFTER 25 US;


   mfi : seq_comp PORT MAP (data, ck, cl_bar, ld_bar, cnt, cnt_out);


END input_output;

FIGURE 9. 53
Test Bench for Testing the standard Configuration of sequential_comparator.

